
○「 資産カーブをチャート上に表示する 」

・アメンボは、以前「擬似トレード提案（その１）」で、擬似トレード方法とチャート上

への、「資産カーブ」表示例を投稿しました。

すると、当然ですが、今度は「実トレード」と「ストラテジー･テスター」のチャート上に

「資産カーブ」を表示してみたくなりました。

・資料調査と試行錯誤の結果、とても簡単な方法に行き着きましたので報告いたします。

本稿では、「資産カーブ」を表示しましたが、「損益」などの他のデータを表示することも

可能です。（当然ですね）

※本稿では、『チャート上への資産カーブ表示』に焦点を絞っていますので、

サンプルとして使用するＥＡは、たまたま検討を始めたばかりの物を使いました、

ご了解ください。

（現状では資産カーブは右肩下がりを続けるＥＡサンプルです）

しかしながら、一応「確認用サンプル」としてダウンロード用に準備しました。

※ＵＳＤＪＰＹ １Ｍ チャート上で実施

※「実トレード」上への「資産カーブ」描画は、「デモ・トレード」画面上で代用しています。

※ダウンロード用資料；

①「myTest_SMA_03.mq4」

②「Disp_Balance.mq4」

-----------------------------------------------------------------------------------

＜資産カーブ表示方法＞

１．ストラテジー・テスター（visual mode）上での表示

（１）ＥＡ「myTest_SMA_03.mq4」を Strategy Tester にセットする

   先ずは、通常と同じバックテストの準備をしておきます。

・通貨ペア； ＵＳＤＪＰＹ

・期間； Ｍ１

・日付と時間を使用 □ にチェック

・Visual Mode □ にチェック

（２）ビジュアル・モードを開始・即ストップしてチャートを開く

・Visual mode の「スライダー」を左の端に置いて、スタート を選択

⇒ Visual mode のチャートが表示されます。

※チャートを観易くするために、「背景を白色、枠線を黒色、ローソク足」に変更する

（３）資産表示インディケータ「Disp_Balance.mq4」を、

ビジュアル・モード（Visual mode）のチャート上にセットする

（４）ビジュアル・モードを進めていく

・Visual Mode のスライダーを少し右の方へ動かす⇒ テスト開始、「資産カーブ」も描画

※テストが更に進行していきます

※［日付と時間を使用］の［終了日］を例えば明日にしておくと、現時点まで達したところで

  ストップ、テスト完了となります。

※このとき、設定されているインディケータをチェックすると下記のようになります。

２．リアルチャート（デモ・モード）上での表示

（１）資産表示インディケータ「Disp_Balance.mq4」を、チャート上にセットする

※この段階では、ＥＡを設定していないので、インディケータは何も表示しません。

※念のため、表示中のインディケータは下記で確認

（２）ＥＡ「myTest_SMA_03.mq4」を Strategy Tester にセットする

※資産カーブが初期化されて、すぐに表示を始めます。

⇒ ＥＡの動作（実行）に従って、資産の変化（カーブ）が表示されていきます

※本稿で使用したＥＡは検討し始めたばかりのものですので、資産カーブは右下がりに

なります。（改良できたところで報告するつもりのＥＡです、ご容赦ください）

本当は、かっこよく、右肩上がりの資産カーブを表示したかったのですが、

今回は間に合わず！

３．MQL4 コード内容一覧

（１）動作確認用ストラテジー＜ＥＡ＞ 「myTest_SMA_03.mq4」
//------------------------------------------------------------------
//           myTest_SMA_03.mq4
//------------------------------------------------------------------
#define myMAGIC  20111207
//
extern double MovingPeriod            = 12;
extern double MovingShift             = 6;
extern double Lots                    = 1;    // Regular
//---------------
int init() {
  return(0);
}
//---------------
int deinit() {
  return(0);
}
//------------------------------------------------------------------
// オープンと待機中ポジションのチェック                                      
//
int CheckCurrentOrders(string symbol)
  {
   int buys=0,sells=0;
//----
   for(int i=0;i<OrdersTotal();i++)
     {
      if(OrderSelect(i,SELECT_BY_POS,MODE_TRADES)==false) break;
      if(OrderSymbol()==Symbol() && OrderMagicNumber()==myMAGIC)
        {
         if(OrderType()==OP_BUY)  buys++;
         if(OrderType()==OP_SELL) sells++;
        }
     }
//---- ﾎﾟｼﾞｼｮﾝ数を返す
   if(buys>0) return(buys);
   else       return(-sells);
  }

//------------------------------------------------------------------
// ﾎﾟｼﾞｼｮﾝを開く（作成）                            
//
void Open_position()
  {
   double ma;
   int    res;
//---- Open 値での動作用処理
//   if(Volume[0]>1) return;
//---- 移動平均値
   ma=iMA(NULL,0,MovingPeriod,MovingShift,MODE_SMA,PRICE_CLOSE,0);
//---- 売り
   if(Open[1]>ma && Close[1]<ma)  
     {
      res=OrderSend(Symbol(),OP_SELL,Lots,Bid,3,0,0,"",myMAGIC,0,Red);
      return;
     }

//---- 買い
   if(Open[1]<ma && Close[1]>ma)  
     {
      res=OrderSend(Symbol(),OP_BUY,Lots,Ask,3,0,0,"",myMAGIC,0,Blue);
      return;
     }
//----
  }

//------------------------------------------------------------------
// ﾎﾟｼﾞｼｮﾝを閉じる                                
//
void Close_position()
  {
   double ma;
//---- Open 値での動作用処理
//   if(Volume[0]>1) return;
//---- 移動平均値
   ma=iMA(NULL,0,MovingPeriod,MovingShift,MODE_SMA,PRICE_CLOSE,0);
//----
   for(int i=0;i<OrdersTotal();i++)
     {
      if(OrderSelect(i,SELECT_BY_POS,MODE_TRADES)==false)        break;
      if(OrderMagicNumber()!=myMAGIC || OrderSymbol()!=Symbol()) continue;
      //---- ｵｰﾀﾞｰ・ﾀｲﾌﾟの確認
      if(OrderType()==OP_BUY)
        {
         if(Open[1]>ma && Close[1]<ma) OrderClose(OrderTicket(),Lots,Bid,3,White);
         break;
        }
      if(OrderType()==OP_SELL)
        {
         if(Open[1]<ma && Close[1]>ma) OrderClose(OrderTicket(),Lots,Ask,3,White);
         break;
        }
     }
//----
  }

//==================================================================                                                 
//
void start()
  {
//---- 資産データをグローバル変数に格納 ----
   GlobalVariableSet("myBalance",AccountBalance());

//---- 足数とトレード可否をチェック
   if(Bars<100 || IsTradeAllowed()==false) return;

//---- 待機中ﾎﾟｼﾞｼｮﾝのチェックと、ﾎﾟｼﾞｼｮﾝ処理
   if(CheckCurrentOrders(Symbol())==0) Open_position();
   else                                Close_position();
//
  }
//------------------------------------------------------------------

（２）資産カーブ表示用＜インディケータ＞「Disp_Balance.mq4」

//
//   myBalance.mq4 
#property indicator_separate_window
#property indicator_buffers 1
#property indicator_color1 Blue
//
double balance[];
//
int init()
  {
   IndicatorShortName("myBalance");  
   SetIndexStyle(0,DRAW_LINE);
   SetIndexBuffer(0,balance);
   SetIndexLabel(0,"Balance");
  }
//
int deinit()
  {
   return(0);
  }
//-----------------------------------------
int start()
  {
//----
   balance[0]=GlobalVariableGet("myBalance");
//----
   return(0);
  }
//-----------------------------------------

※解説

＜ＥＡ＞

資産カーブを描画するために必要な記述は、下記の「青書」部分（一行）のみです。

   GlobalVariableSet("myBalance",AccountBalance());

＜インディケータ＞

ポイントは、下記の「一行」のみです。

  balance[0]=GlobalVariableGet("myBalance");

以 上


